use "/homeKRTK/health_shocks/data_for_main_regs_event_study_20231203.dta", clear
drop if control==1
append using "/homeKRTK/health_shocks/additional_data_for_all_regs_controls_nod_20240422.dta"
append using "/homeKRTK/health_shocks/additional_data_for_IV_and_placebo_regs_20240416.dta"

replace treated=1 if d==2

gen NONEMP = 1 if ABS==1
	recode NONEMP (.=0) if ABS==0
	
set seed 12345

keep if d<=6
*keep if d>=3 & d<=6
*keep if NONEMP !=.

gen month_shock = t-12*int(t/12)
	recode month_shock (0=12)
gen month_return = month_shock+d

*keep if treated == 1 | control == 1

forval i=1/8 {
	local j=`i'+8
	rename mean_fq`i'_w w_q`j'
	replace mean_fq`i'_FEj_imp =. if flag_missingFEj_fq`i'==1
	rename mean_fq`i'_FEj_imp FE_q`j'


}
forval i=1/8 {
	local j=8-`i'
	rename mean_lq`i'_w w_q`j'
	replace mean_lq`i'_FEj_imp =. if flag_missingFEj_lq`i'==1
	rename mean_lq`i'_FEj_imp FE_q`j'
}


* reshape *
capture drop FEj_t4 FEj_t5 FEj_t6 FEj_t7 FEj_t8 FEj_t9 FEj_t10 FEj_t11 FEj_t12 
capture drop lnw_restr_t4 lnw_restr_t5 lnw_restr_t6 lnw_restr_t7 lnw_restr_t8 lnw_restr_t9 lnw_restr_t10 lnw_restr_t11 lnw_restr_t12 

reshape long w_q FE_q, i(anon) j(relative_time)


recode teaor1_h1 (-1=99) (.=100)

cap drop date
gen date=t+(relative_time-8)*3
replace date=t+d+(relative_time-9)*3 if relative_time>=8

rename kor kor_orig

rename ev ev_orig
rename t t_orig
rename date t

gen ev=floor((t-1)/12)+2003
merge 1:1 anon t using "/homeProspSSD/Admin3/admin3_alap.dta", nogen keep(master match) keepusing(vallazon1 wtip1 wh1 feor1_h2 kor)
replace wh1=. if wh1>40
rename vallazon1 vallazon
merge m:1 vallazon ev using "/homeKRTK/health_shocks/firmquality.dta", nogen keep(master match)

merge 1:1 anon t using "/homePCIESSD800/Admin3/Output_data/Alap/admin3_passziv_kieg_H1_v1.dta", nogen keepusing(passziv_ho)
replace passziv_ho=0 if passziv_ho==.

by anon, sort: gegen full_d=min(d)
drop if full_d==. //drop anons who are never in our baseline sample

rename t date
rename t_orig t

by anon, sort: gegen full_t=min(t)

gen sl_relevant1_=passziv_ho if date>=full_t & date<=full_t+1
by anon, sort: gegen sl_relevant1=total(sl_relevant1_)
drop sl_relevant1_

drop if d==. //back to baseline+falsification sample
drop if d>6

keep if sl_relevant1>0 | control==1 //sample restriction!!!

keep anon t d control

duplicates drop
gen insample=1

save "/homeKRTK/health_shocks/EventStudy/EventStudy_sample.dta", replace



use "/homeKRTK/health_shocks/data_for_futures_regs_with_d_2_6_20240508.dta", clear
merge m:1 anon using "/homeKRTK/health_shocks/EventStudy/EventStudy_sample.dta", nogen keepusing(insample) keep(master match)
keep if insample==1
keep if ev<=2014
 
gen month_shock = t-12*int(t/12)
	recode month_shock (0=12)
gen month_return = month_shock+d

keep if treated == 1 | control == 1

forval i=1/8 {
	local j=`i'+8
	rename mean_fq`i'_w w_q`j'
	replace mean_fq`i'_FEj_imp =. if flag_missingFEj_fq`i'==1
	rename mean_fq`i'_FEj_imp FE_q`j'


}
forval i=1/8 {
	local j=8-`i'
	rename mean_lq`i'_w w_q`j'
	replace mean_lq`i'_FEj_imp =. if flag_missingFEj_lq`i'==1
	rename mean_lq`i'_FEj_imp FE_q`j'
}


* reshape *
capture drop FEj_t4 FEj_t5 FEj_t6 FEj_t7 FEj_t8 FEj_t9 FEj_t10 FEj_t11 FEj_t12 
capture drop lnw_restr_t4 lnw_restr_t5 lnw_restr_t6 lnw_restr_t7 lnw_restr_t8 lnw_restr_t9 lnw_restr_t10 lnw_restr_t11 lnw_restr_t12 

reshape long w_q FE_q, i(anon t) j(relative_time)

cap drop post
gen post=1 if relative_time~=4
replace post=0 if relative_time==4

cap drop interac
gen interac=post*treated

recode teaor1_h1 (-1=99) (.=100)

gen date=t+(relative_time-8)*3
replace date=t+d+(relative_time-9)*3 if relative_time>=8



cap estimates clear

egen group_id=group(anon t)
xtset group_id relative_time

cap drop rt
gen rt=relative_time
replace rt=7 if treated==0

cap drop rtdum*
tabulate rt, generate(rtdum)

forval i=2/8 {
	replace rtdum`i'=rtdum`i'-rtdum1
}


winsor2 w_q, cuts(0 99) by(ev)
winsor2 FE_q, cuts(0 99) by(ev)

rename t t_orig
rename date t
rename ev ev_orig

gen ev=floor((t-1)/12)+2003
merge m:1 anon t using "/homeProspSSD/Admin3/admin3_alap.dta", nogen keep(master match) keepusing(vallazon1)
rename vallazon1 vallazon
merge m:1 vallazon ev using "/homeProspSSD/Admin3/admin3_nav_eredeti.dta", nogen keep(master match) keepusing(letszam)

rename t date
rename t_orig t

duplicates drop anon date, force //get rid of redundant controls

***Input for Figures A3 & B6 - saved in "$folder\EventStudy_Marc2025.xlsx"


cap erase /homeKRTK/health_shocks/EventStudy/log_EventStudy_futures.log
log using /homeKRTK/health_shocks/EventStudy/log_EventStudy_futures.log, append


areg w_q_w i.date rtdum2-rtdum16 if ev_orig>=2010 & ev_orig<=2014 & letszam~=. & FE_q_w~=., absorb(anon) rob
lincom -rtdum2-rtdum3-rtdum4-rtdum5-rtdum6-rtdum7-rtdum8
areg FE_q_w i.date rtdum2-rtdum16 if ev_orig>=2010 & ev_orig<=2014 & letszam~=. & w_q_w~=., absorb(anon) rob
lincom -rtdum2-rtdum3-rtdum4-rtdum5-rtdum6-rtdum7-rtdum8

areg w_q_w i.date rtdum2-rtdum16 if ev_orig>=2012 & ev_orig<=2014 & letszam~=. & FE_q_w~=., absorb(anon) rob
lincom -rtdum2-rtdum3-rtdum4-rtdum5-rtdum6-rtdum7-rtdum8
areg FE_q_w i.date rtdum2-rtdum16 if ev_orig>=2012 & ev_orig<=2014 & letszam~=. & w_q_w~=., absorb(anon) rob
lincom -rtdum2-rtdum3-rtdum4-rtdum5-rtdum6-rtdum7-rtdum8

log close

